
$oninline 
/* 
 GAMS program used to estimate output oriented technical efficiency 
with nonincreasing returns to scale. 
 
 Author:  John B. Walden 
          NMFS/NEFSC 
          166 Water St.  
          Woods Hole, MA  02543 
          (508) 495-2355 
          John.Walden@Noaa.Gov 
 
This version will estimate technical efficiency for 
each vessel, based on an output oriented DEA model. 
 
*/ 
 
/* The following line turns off listing of 
some elements in the GAMS listing file*/ 
 
$OFFSYMLIST OFFSYMXREF OFFUELLIST OFFUELXREF 
 
/*NEXT DEFINE INPUTS AND OUTPUTS*/ 
 
SET INOUT /y1*y3, x1*x5/ 
    OUTPUT(INOUT) /y1*y3/ 
    INPUT(INOUT) /x1*x5/ 
 
 
    OBS /1*200/ 
    SUBOBS(OBS) /1*50/ 
    ACTOBS(OBS) 
 
/*We have allocated enough memory for 200 observations, but our 
data set only contains 50 observations (subobs)*/ 
 
 
/*Next, define an alias for the set SUBOBS */    
     
alias (subobs, subobs1)     
 
/* The include statement below reads in an external data 
file which contains a table of observations, inputs 
and outputs.   
 
The offlisting command means that the data won't 
be included in the listing file. 
 
*/ 
 
$OFFLISTING 
 
TABLE ACT(OBS,INOUT)  INPUT OUTPUT TABLE 
$ondelim 
$INCLUDE "data_test.csv" 
$offdelim 



 
$ONLISTING 
 
VARIABLES 
 
 theta      efficiency score 
 weight(obs)  weights; 
  
 POSITIVE Variable weight; 
  
 EQUATIONS 
   CONSTR1(OUTPUT, OBS)  DEA constraint for each output 
   CONSTR2(INPUT, OBS)  DEA constraint for each input 
   CONSTR3    DEA constraint for imposing VRS;  
 
   CONSTR1(OUTPUT, ACTOBS).. SUM(SUBOBS, WEIGHT(SUBOBS)*ACT(SUBOBS,OUTPUT)) =G= 
        THETA*ACT(ACTOBS, OUTPUT); 
    
   CONSTR2(INPUT, ACTOBS).. SUM(SUBOBS, WEIGHT(SUBOBS)*ACT(SUBOBS,INPUT)) =L= 
         ACT(ACTOBS, INPUT); 
  
   CONSTR3.. SUM(SUBOBS, WEIGHT(SUBOBS)) =L= 1; 
 
/*Define a parameter to hold results for each pass through the loop*/ 
 
    
 PARAMETER 
  
   score1(obs)  efficiency scores 
; 
 
/*Define an external file to hold results which tell whether model solved 
at each iteration*/ 
 
file primal /teout_res.txt/; 
 
/*The model defined below consists of three equations. 
CONSTR1, CONSTR2, CONSTR3. We use the term ALL in the model statement 
rather than listing the equations separately as we did in the 
CRS example */ 
 
 MODEL TEDEA /ALL/; 
  
tedea.solprint=2;         /*Turn off writing results to solution file*/ 
tedea.solvelink=2;       /*Keep model in memory. Improves solution time*/ 
  
 LOOP(SUBOBS1, 
       
      
      ACTOBS(OBS)=NO; 
      ACTOBS(SUBOBS1)=YES; 
 
      SOLVE TEDEA maximizing THETA USING LP; 
     
       
      score1(SUBOBS1) = theta.l; 



 
      put primal; 
       
      if ((tedea.modelstat eq 1 and  tedea.solvestat eq 1), 
 
        put @1, subobs1.tl, @10, "optimal", @20, "normal completion" / 
 
      else 
 
        put @1, subobs1.tl, @10, tedea.modelstat:>2:0, 
             @20, tedea.solvestat:>2:0/ 
      );    
 
   ); 
       
 
/*The next file is to output results to a file to be imported 
into a spreadsheet program.  Results could also be printed to 
the listing file with the use of the display command*/ 
 
file res /te_out_nrs.csv/ ; 
 
res.pc=5; 
 
put res; 
 
loop (subobs1, 
 
   put subobs1.tl, score1(subobs1)/ 
   ); 
putclose; 
  
 


